Dowling21.wxmx: Optimal Control Theory

TABLE OF CONTENTS


```
(%i2) load(draw)$ set_draw_defaults(line_width=2, grid = [2,2], point_type = filled_circle,
        head_type = 'nofilled, head_angle = 20, head_length = 0.5,
          background_color = light_gray, draw_realpart=false)$

(%i3) load ("Econ2.mac");
(%o3) c:/work5/Econ2.mac
```

1 Preface

Dowling21.wxmx is one of a number of wxMaxima files available in the section
Economic Analysis with Maxima
on my CSULB webpage.

In Dowling21.wxmx, we use Maxima to discuss the maximization of an integral over a finite period of time containing both a time dependent state variable x(t) subject to constraints and a time dependent control variable y(t) whose forms are sought using the calculus of variations, following Dowling's Chapter 21: "Optimal Control Theory". The problems considered are restricted to the context of continuous time, a finite time horizon, and fixed endpoint constraints on the state variable.

We have changed some of the symbols used in particular problems. We use p(t) to represent the co-state variable instead of Dowling's λ(t). An approximate pdf translation (using Microsoft print to pdf) is available as Dowling21fit.pdf. That pdf file can be searched using Ctrl-F.

Ted Woollett
https://home.csulb.edu/~woollett/
woollett@charter.net
April 7, 2022

2 References

Introduction to Mathematical Economics, 3rd ed., Edward T. Dowling, 2012, Schaum's Outline Series, McGraw-Hill.

Fundamental Methods of Mathematical Economics, Alpha C. Chiang and Kevin Wainwright, 4th ed., 2005, McGraw-Hill

Elements of Dynamic Optimization, Alpha C. Chiang, 2012, Waveland Press

3 *Terminology and Statement of the Problem*

Roughly quoting Dowling:
"In optimal control theory, the aim is to find the optimal time path for a 'control variable', which we shall denote as y. The variable for which we previously sought an optimal time path in the calculus of variations [Ch. 20], known as a 'state variable', we shall continue to designate as x. The goal of optimal control theory is to select a stream of values over time for the control variable that will optimize a functional (J) subject to the constraints set on the state variable."

"Optimal control theory problems involving 1) continuous time, 2) a finite time horizon, and 3) fixed endpoints are generally written:

Maximize the functional: J = integrate (f[x(t), y(t), t], t, 0, T)

subject to: dx/dt = g[x(t), y(t), t], x(0) = x0, x(T) = xT, (21.1)

where J = the value of the functional to be optimized; y(t) = the control variable, so called because its value is selected or 'controlled' to optimize J; x(t) = the state variable, which changes over time according to the differential equation implied by the constraint dx/dt = g[x(t), y(t), t] , and whose value is 'controlled' by the presence of the control variable y(t) in that constraint; and
t = time. The solution to an optimal control problem demarcates the optimal dynamic time path for the control variable y(t)."

4 *Necessary Conditions for Max. with Fixed End Pts.*

"Dynamic optimization of a functional subject to a constraint on the state variable in optimal control involves a 'Hamiltonian function H' similar to the Lagrangian function in concave programming.

In terms of (21.1) the Hamiltonian is defined as"

$$H[x(t), y(t), p(t), t] = f[x(t), y(t),t] + p(t)*g[x(t), y(t), t] \quad (21.2)$$

where p(t) is called the 'co-state variable' [also the 'generalized momentum', the 'marginal valuation' of x(t), the 'shadow price', and 'how much a unit increment in x at time t contributes to the optimal objective functional J' ".

Dowlingl uses λ(t) instead of p(t) for the co-state variable. In a problem in which the state variable is a price, we can use symbols p(t) as the state variable and λ(t) as the co-state variable as a more natural notation.

"Similar to the Lagrangian multiplier, the co-state variable p(t) estimates the marginal value or shadow price of the associated state variable x(t)."

"Assuming the Hamiltonian is differentiable in y and strictly concave, so there is an interior solution and not an endpoint solution, the necessary conditions for maximization are"

1. $\partial H/\partial y = 0$

2. a) $dp/dt = -\partial H/\partial x$ b) $dx/dt = \partial H/\partial p$

3. a) $x(0) = x0$ b) $x(T) = xT$

"The first two conditions are known as 'the [Pontryagin] maximum principle' (PMP) and the third is called the 'boundary condition.' The two equations of motion in the second condition are generally referred to as the 'Hamiltonian system' or the 'canonical system'."

"For minimization (instead of maximization) the objective functional can simply be multiplied by -1, as in concave programming. If the solution involves an end point, ∂H/∂y need not equal zero in the first condition, but H must still be maximized with respect to y. See Ch. 13, Ex. 9 and Fig. 13-1(b)-(c) for clarification."

5 *Sufficient Conditions for Maximization*

Quoting Dowling:
"Assuming the maximum principle representing the necessary conditions for maximization in optimal control are satisfied, the sufficiency conditions will be fulfilled if

1. Both the objective functional [integrand] f(x,y,t) and the constraint g(x,y,t) are differentiable and jointly concave in x and y, and

2. p(t) >= 0, if the constraint g(x,y,t) is nonlinear in x or y. If the constraint is linear, then p(t) may assume any sign."

"Linear functions are always both concave and convex, but neither strictly concave nor strictly convex. For nonlinear functions, an easy test for joint concavity is the simple discriminant test.

Given the discriminant matrix D of the second order derivatives of a function [say, f(x,y)]:
D = matrix ([fxx, fxy], [fyx, fyy]),
a function f(x,y) will be 'strictly concave' if the discriminant is 'negative definite':
D1 = D[1,1] = fxx < 0 and D2 = determinant(D) > 0
and 'simply concave' if the discriminant is 'negative semidefinite':
D1 = fxx <= 0 and D2 >= 0.

A negative definite discriminant indicates a global maximum and is, therefore, always sufficient for a maximum. A negative semidefinite discriminant is indicative of a relative maximum and is sufficient for a maximum if the test is conducted for every possible ordering of the variables (x,y) with similar results."

6 *ConcaveTest(f, x, y)*

The Maxima function ConcaveTest (f, x, y), defined in Econ2.mac, determines if f(x,y) is jointly concave in x and y using discriminant tests.

The matrix d1 which is displayed is equivalent to
matrix ([fxx, fxy], [fyx, fyy]). d11 is fxx and dd1 is determinant(d1).

The matrix d2 which is displayed is equivalent to matrix ([fyy, fyx], [fxy, fxx]). d21 is fyy and dd2 is determinant (d2).

(%i4) ConcaveTest (4*x - 5*y^2, x, y);

$$d1 = \begin{pmatrix} 0 & 0 \\ 0 & -10 \end{pmatrix}$$

$d11 = 0$

$dd1 = 0$

$$d2 = \begin{pmatrix} -10 & 0 \\ 0 & 0 \end{pmatrix}$$

$d21 = -10$

$dd2 = 0$

simply concave

(%o4) *done*

(%i5) ConcaveTest (8*y, x, y)$

$$d1 = \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

$d11 = 0$

$dd1 = 0$

$$d2 = \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

$d21 = 0$

$dd2 = 0$

simply concave

7 Problems with Fixed End Points

7.1 Example 1

Use the necessary conditions for maximization to solve the problem:

Maximize the functional: J = integrate (4*x - 5*y^2, t, 0, 3)

subject to: dx/dt = 8*y, x(0) = 2, and x(3) = 117.2.

7.1.1 Interior Solution equations of motion

A. Set up the Hamiltonian H + p*g:

(%i6) H : 4*x - 5*y^2 + p*8*y;

(H) $-5\,y^2 + 8\,p\,y + 4\,x$

B. Assuming an interior solution, apply the maximum principle (1 & 2):
1. ∂H/∂y = 0:

(%i7) dy : diff (H, y);

(dy) $8\,p-10\,y$

(%i8) sy : solve(dy,y);

(sy) $[y=\frac{4\,p}{5}]$

Let sy stand for y(t) in terms of p(t):

(%i9) sy : rhs (sy[1]);

(sy) $\frac{4\,p}{5}$

so y(t) = 4*p(t)/5.

2. use dx/dt = diff(H,p) and dp/dt = -diff(H,x).

Let xd stand for dx/dt = ∂H/∂p.

(%i10) xd : diff(H,p);

(xd) $8\,y$

Make use of our result that sy stands for 4*p/5.

(%i11) xd : at (xd, y = sy);

(xd) $\frac{32\,p}{5}$

So dx/dt = 32*p/5, which means to find x(t) we need to know p(t).
Let pd stand for dp/dt.

(%i12) pd : - diff (H, x);

(pd) -4

We can immediately integrate this first order ode:
dp/dt = -4,
to get p(t) in terms of an arbitrary constant k1.

Let sp stand for p(t) in terms of the constant k1.

```
(%i13) sp : integrate(pd, t) + k1;
```
(sp) $k1-4\,t$

Now replace p(t) in expression dx/dt (xd).

```
(%i14) xd : at(xd, p = sp), expand;
```
(xd) $\frac{32\,k1}{5}-\frac{128\,t}{5}$

We can then integrate dx/dt = F(t) in terms of another arbitrary integration constant k2.
Let sx stand for x(t) in terms of k1 and k2.

```
(%i15) sx : integrate (xd, t) + k2;
```
(sx) $-\frac{64\,t^2}{5}+\frac{32\,k1\,t}{5}+k2$

Apply the given boundary condtions at t = 0 and t = 3 to determine the values of k1 and k2.

```
(%i16) ksoln : solve ([ at(sx, t = 0) = 2, at (sx, t = 3) = 117.2] );
```
(ksoln) $[[k1=12,k2=2]]$

```
(%i17) ksoln : ksoln[1];
```
(ksoln) $[k1=12,k2=2]$

Update sx, which is x(t) to reflect the consequences of the endpoint boundary conditions.

```
(%i18) sx : at (sx, ksoln );
```
(sx) $-\frac{64\,t^2}{5}+\frac{384\,t}{5}+2$

Obviously, x(0) = 2.

```
(%i19) at (sx, t = 0);
```
(%o19) 2

```
(%i20) at (sx, t = 3), numer;
```
(%o20) 117.2

Update sp which is p(t), the co-state variable, to reflect the consequences of the boundary conditions:

```
(%i21) sp : at(sp, ksoln );
(sp)   12−4 t
```

Obviously, p(0) = 12, p(3) = 0.

```
(%i22) at(sp, t = 3);
(%o22) 0
```

sy is y(t), the control variable, which is proportional to p. Replace p with sp.

```
(%i23) sy : at (sy, p = sp), expand;
```

(sy) $\frac{48}{5}-\frac{16\,t}{5}$

The optimal path of the control variable y(t) is linear, starting at (0, 9.6) and ending at (3,0), with a slope of -16/5 = - 3.2.

Value of y(0) in terms of a floating point number:

```
(%i24) at(sy, t = 0), numer;
(%o24) 9.6
```

Value of y(3) in terms of a floating point number. Maxima's numerical calculations carry along 16 digits, so the following answer is approx. 0.

```
(%i25) at(sy, t = 3), numer;
```

(%o25) $-1.7764\ 10^{-15}$

Doing the evaluation symbolically, we get exactly 0.

```
(%i26) at(sy, t = 3);
(%o26) 0
```

So y(0) = 9.6, y(3) = 0.

7.1.2 Plot of solutions of Example 1

(%i27) wxdraw2d (xlabel = "t", key = " state var x(t)", yrange = [0, 140], key_pos = top_left, explicit (sx, t, 0, 3), color = red, key = "control var y(t)", explicit (sy, t, 0, 3), color = brown, key = " co-state var p(t)", explicit (sp, t, 0, 3))$

(%t27)



If we relace x,y,and p with our solutions x(t), y(t) and p(t) we can evaluate H(t):

(%i28) Ht : at (H, [x = sx, y = sy, p = sp]), expand;

(Ht) $\frac{2344}{5}$

We evaluate the extremal value of J = integrate(f, t, 0, 3), all we need are the x(t) and y(t) solutions. We first define f as an expression in terms of x and y, and then replace x and y by sx and sy, respectively [redefining f as an expression depending only on t].

(%i30) f : 4*x - 5*y^2;
f : at (f, [x = sx, y = sy]), expand;

(f) $4\,x-5\,y^2$

(f) $-\frac{512\,t^2}{5}+\frac{3072\,t}{5}-\frac{2264}{5}$

Integrating this expression over the time interval [0, 3] then gives us the extremum value of J.

(%i31) Jextr : integrate (f, t, 0, 3);

(Jextr) $\frac{2424}{5}$

(%i32) float (%);

(%o32) 484.8

7.1.3 Sufficient Conditions for Example 1

Check the sufficient conditions for a maximum in J.

1. The constraint function g = 8*y is linear in y and is thus always both concave and convex. That implies we don't need to worry about the sign of p(t), the co-state variable.

2. We need the integrand f(x,y) of the functional J to be jointly concave in x and y.

ConcaveTest (f, x, y), discussed above, finds that f(x,y) is jointly (simply) concave in both x and y, and the maximum is therefore a relative maximum.

```
(%i33) ConcaveTest (4*x - 5*y^2, x, y)$
```

$$d1=\begin{pmatrix}0 & 0\\ 0 & -10\end{pmatrix}$$

$d11=0$

$dd1=0$

$$d2=\begin{pmatrix}-10 & 0\\ 0 & 0\end{pmatrix}$$

$d21=-10$

$dd2=0$

simply concave

7.2 Prob. 21.1

Maximize the functional: J = integrate (6*x - 4*y^2, t, 0, 2)
subject to: dx/dt = 16*y, x(0) = 24, x(2) = 408.

A. Set up the Hamiltonian, H = f + p*g.

```
(%i34) H : 6*x - 4*y^2 + p*16*y;
```

(H) $-4\,y^2+16\,p\,y+6\,x$

B. Assuming an interior solution, apply the maximum principle (1 & 2):
1. ∂H/∂y = 0:

```
(%i35) dy : diff (H, y);
```

(dy) $16\,p-8\,y$

```
(%i36) sy : solve(dy,y);
```

(sy) $[y=2\,p]$

(%i37) sy : rhs (sy[1]);

(sy) $2\,p$

so y(t) = 2*p(t)

2. use dx/dt = diff(H,p) and dp/dt = -diff(H,x).

(%i38) xd : diff(H,p);

(xd) $16\,y$

(%i39) xd : at (xd, y = sy);

(xd) $32\,p$

dx/dt depends on p(t).

(%i40) pd : - diff (H, x);

(pd) -6

We can immediately integrate this first order ode:
dp/dt = - 6,
to get p(t) in terms of an arbitrary constant k1.

sp is p(t) in terms of constant k1.

(%i41) sp : integrate(pd, t) + k1;

(sp) $k1-6\,t$

Replace p(t) in the expression xd (which stands for dx/dt)

(%i42) xd : at(xd, p = sp), expand;

(xd) $32\,k1-192\,t$

We can then integrate dx/dt = F(t) in terms of another arbitrary integration constant k2.
Let sx stand for x(t) in terms of k1 and k2.

(%i43) sx : integrate (xd, t) + k2;

(sx) $-96\,t^2+32\,k1\,t+k2$

Apply the boundary condtions at t = 0 and t = 3 to determine the values of k1 and k2.

(%i44) ksoln : solve ([at(sx, t = 0) = 24, at (sx, t = 2) = 408]);

(ksoln) $[[k1=12,k2=24]]$

```
(%i45) ksoln : ksoln[1];
(ksoln) [k1=12,k2=24]
```

Update sx, which is x(t).

```
(%i46) sx : at (sx, ksoln );
```

(sx) $-96\,t^2+384\,t+24$

Check x(0) = 24, x(2) = 408.

```
(%i47) [ at (sx, t = 0), at (sx, t = 2)];
(%o47) [24,408]
```

Update sp which is p(t), the co-state variable.

```
(%i48) sp : at(sp, ksoln );
```

(sp) $12-6\,t$

We now find that p(0) = 12, p(2) = 0.

```
(%i49) [at (sp, t = 0),  at(sp, t = 2) ];
(%o49) [12,0]
```

sy is y(t), the control variable, which is proportional to p, so update sy:

```
(%i50) sy : at (sy, p = sp), expand;
```

(sy) $24-12\,t$

The optimal path of the control variable y(t) is linear, starting at (0, 24) and ending at (2,0), with a slope of -12.

Check initial and final values of the control variable y(t):

```
(%i51) [ at(sy, t = 0), at(sy, t = 2)];
(%o51) [24,0]
```

Value of y(2).

(%i52)
```
wxdraw2d (xlabel = "t", key = " state var x(t)", yrange = [0, 450], key_pos = top_left,
    explicit (sx, t, 0, 2), color = red, key = "control var y(t)", explicit (sy, t, 0, 2),
    color = brown, key = " co-state var p(t)", explicit (sp, t, 0, 2))$
```

(%t52)



Check the sufficient conditions for a maximum in J for Prob. 21.1.

1. The constraint function g = 16*y is linear in y and is always both concave and convex, and we hence don't need to worry about the sign of p(t), the co-state variable.

2. We need the integrand f(x,y) of the functional J to be jointly concave in x and y. ConcaveTest (f, x, y) finds f(x,y) is jointly (simply) concave in both x and y, and the maximum is therefore a relative maximum. (Note: f above ended up as an expression depending only of t.)

(%i53)
```
ConcaveTest (6*x - 4*y^2, x, y)$
```

$$d1=\begin{pmatrix}0 & 0\\ 0 & -8\end{pmatrix}$$

$d11=0$

$dd1=0$

$$d2=\begin{pmatrix}-8 & 0\\ 0 & 0\end{pmatrix}$$

$d21=-8$

$dd2=0$

simply concave

7.3 Prob. 21.2

Maximize the functional: J = integrate (5*x + 3*y - 2*y^2, t, 0, 1)
subject to: dx/dt = 6*y, x(0) = 7, x(1) = 70.

A. Set up the Hamiltonian, H = f + p*g.

(%i54) H : 5*x + 3*y - 2*y^2 + p*6*y;
(H) $-2\,y^2+6\,p\,y+3\,y+5\,x$

B. Assuming an interior solution, apply the maximum principle (1 & 2):
1. ∂H/∂y = 0:

(%i55) dy : diff (H, y);
(dy) $-4\,y+6\,p+3$

(%i56) sy : solve(dy,y), expand;
(sy) $[y=\frac{3\,p}{2}+\frac{3}{4}]$

sy is y(t) solution in terms of p(t).

(%i57) sy : rhs (sy[1]);
(sy) $\frac{3\,p}{2}+\frac{3}{4}$

2. Require: dx/dt = diff(H,p) and dp/dt = -diff(H,x).

(%i58) xd : diff(H,p);
(xd) $6\,y$

(%i59) xd : at (xd, y = sy), expand;
(xd) $9\,p+\frac{9}{2}$

dx/dt depends on p(t).

(%i60) pd : - diff (H, x);
(pd) -5

We can immediately integrate this first order ode:
dp/dt = -5,
to get p(t) in terms of an arbitrary constant k1.

sp is p(t) in terms of the constant k1.

```
(%i61) sp : integrate(pd, t) + k1;
```
(sp) $k1-5\,t$

replace p(t) in expression xd, which stands for dx/dt.

```
(%i62) xd : at(xd, p = sp), expand;
```
(xd) $-45\,t+9\,k1+\frac{9}{2}$

We can then integrate dx/dt = F(t) in terms of another arbitrary integration constant k2.
sx is x(t) in terms of k1 and k2.

```
(%i63) sx : integrate (xd, t) + k2;
```
(sx) $-\frac{45\,t^2}{2}+9\,k1\,t+\frac{9\,t}{2}+k2$

Apply the boundary condtions at t = 0 and t = 1 to determine the values of k1 and k2.

```
(%i64) ksoln : solve ([ at(sx, t = 0) = 7, at (sx, t = 1) = 70] );
```
(ksoln) $[[k1=9,k2=7]]$

```
(%i65) ksoln : ksoln[1];
```
(ksoln) $[k1=9,k2=7]$

Update sx, which is x(t).

```
(%i66) sx : at (sx, ksoln );
```
(sx) $-\frac{45\,t^2}{2}+\frac{171\,t}{2}+7$

```
(%i67) float (sx);
```
(%o67) $-22.5\,t^2+85.5\,t+7.0$

```
(%i68) [at(sx, t = 0), at (sx, t = 1) ];
```
(%o68) $[7,70]$

Update sp which is p(t), the co-state variable.

```
(%i69) sp : at(sp, ksoln );
(sp)   9-5 t

(%i70) [at (sp, t = 0), at(sp, t = 1) ];
(%o70) [9,4]
```

sy stands for y(t), the control variable, which is proportional to p, so update sy:

```
(%i71) sy : at (sy, p = sp), expand;
```

(sy) $\frac{57}{4}-\frac{15\,t}{2}$

```
(%i72) float (sy);
(%o72) 14.25-7.5 t

(%i73) [at (sy, t = 0),  at(sy, t = 1)], numer;
(%o73) [14.25,6.75]
```

The optimal path of the control variable y(t) is linear, starting at (0, 14.25) and ending at (1, 6.75), with a slope of - 7.5.

```
(%i74) wxdraw2d (xlabel = "t", key = " state var x(t)", yrange = [0, 100], key_pos = top_left,
    explicit (sx, t, 0, 1), color = red, key = "control var y(t)", explicit (sy, t, 0, 1),
    color = brown, key = " co-state var p(t)", explicit (sp, t, 0, 1))$
```

(%t74)



Check the sufficient conditions for a maximum in J for Prob. 21.2.

1. The constraint function g = 6*y is linear in y and is always both concave and convex, and we hence don't need to worry about the sign of p(t), the co-state variable.

2. We need the integrand f(x,y) of the functional J to be jointly concave in x and y. ConcaveTest (f, x, y) finds f(x,y) is jointly (simply) concave in both x and y, and the maximum is therefore a relative maximum.

(%i75) ConcaveTest (5*x + 3*y - 2*y^2, x, y)$

$$d1=\begin{pmatrix}0 & 0\\ 0 & -4\end{pmatrix}$$

$d11=0$

$dd1=0$

$$d2=\begin{pmatrix}-4 & 0\\ 0 & 0\end{pmatrix}$$

$d21=-4$

$dd2=0$

simply concave

8 Neccessary Conditions for Max. with a Free End Point

Quoting Dowling:
"The general format for an optimal control problem involving continuous time with a finite time horizon and a free endpoint is:

Maximize the functional J = integrate (f[x(t), y(t), t], t, 0, T)
subject to: dx/dt = g[x(t), y(t), t],
x(0) = x0, x(T) = free and unrestricted.

"Assuming an interior solution, the first two conditions for maximization, comprising the maximum condition, remain the same, but the third or boundary condition changes."

The Hamiltonian function is: H[x(t), y(t), p(t), t] = f[x(t), y(t),t] + p(t)*g[x(t), y(t), t] with p(t) the co-state variable, x(t) the state variable, y(t) the control variable.

"Assuming the Hamiltonian is differentiable in y and strictly concave so there is an interior solution and not an endpoint solution, the necessary conditions for maximization are

1. ∂H/∂y = 0

2. a) dp/dt = - ∂H/∂x b) ∂x/∂t = ∂H/∂p

3. a) x(0) = x0 b) p(T) = 0

"The very last condition, p(T) = 0 is known as the 'transversality condition' for a free endpoint. The rationale for the transversality condition follows straightforward from what we learned in concave programming. If the value of x at T is free to vary, the constraint must be nonbinding and the shadow price p evaluated at T must equal 0, i.e., p(T) = 0. See Prob. 21.4 to 21.6."

"For minimization (instead of maximization), the objective functional can simply be multiplied by -1, as in concave programming. If the solution involves an end point, ∂H/∂y need not equal zero in the first condition, but H must still be maximized with respect to y."

8.1 Example 3

Maximize the functional J = integrate (3*x - 2*y^2, t, 0, 2)
subject to: dx/dt = 8*y, x(0) = 5, x(2) = free and unrestricted.

A. Set up the Hamiltonian, H = f + p*g.

```
(%i76) H : 3*x - 2*y^2 + p*8*y;
```

(H) $-2\,y^2+8\,p\,y+3\,x$

B. Assuming an interior solution, apply the maximum principle (1 & 2):
1. ∂H/∂y = 0:

```
(%i77) dy : diff (H, y);
```

(dy) $8\,p-4\,y$

```
(%i78) sy : solve(dy,y);
```

(sy) $[y=2\,p]$

sy is y(t) solution in terms of p(t).

```
(%i79) sy : rhs (sy[1]);
(sy)   2 p
```

2. use dx/dt = diff(H,p) and dp/dt = -diff(H,x).

```
(%i80) xd : diff(H,p);
(xd)   8 y
```

```
(%i81) xd : at (xd, y = sy);
(xd)   16 p
```

dx/dt depends on p(t).

```
(%i82) pd : - diff (H, x);
(pd)   −3
```

We can immediately integrate this first order ode:
dp/dt = -3,
to get p(t) in terms of an arbitrary constant k1.

sp is p(t) in terms of constant k1.

```
(%i83) sp : integrate(pd, t) + k1;
(sp)   k1−3 t
```

replace p(t) in expression dx/dt (xd).

```
(%i84) xd : at(xd, p = sp), expand;
(xd)   16 k1−48 t
```

We can then integrate dx/dt = F(t) in terms of another arbitrary integration constant k2.
sx is x(t) in terms of k1 and k2.

```
(%i85) sx : integrate (xd, t) + k2;
```

(sx) $-24\, t^2+16\, k1\, t+k2$

Apply the boundary condtions at t = 0 and t = 2 to determine the values of k1 and k2.
Instead of x(2) value we use p(2) value (transversality condition).

(%i86) ksoln : solve ([at(sx, t = 0) = 5, at (sp, t = 2) = 0]);

(ksoln) $[[k1=6, k2=5]]$

(%i87) ksoln : ksoln[1];

(ksoln) $[k1=6, k2=5]$

Update sx, which is x(t).

(%i88) sx : at (sx, ksoln);

(sx) $-24\,t^2+96\,t+5$

(%i89) [at (sx, t = 0), at (sx, t = 2)];

(%o89) $[5, 101]$

Update sp which is p(t), the co-state variable.

(%i90) sp : at(sp, ksoln);

(sp) $6-3\,t$

p(0) = 6, p(2) = 0.

(%i91) [at (sp, t = 0), at(sp, t = 2)];

(%o91) $[6, 0]$

sy is y(t), the control variable, which is proportional to p, so update sy:

(%i92) sy : at (sy, p = sp), expand;

(sy) $12-6\,t$

The optimal path of the control variable y(t) is linear, starting at (0, 12) and ending at (2, 0), with a slope of - 6.

(%i93) wxdraw2d (xlabel = "t", key = " state var x(t)", yrange = [0, 120], key_pos = top_left, explicit (sx, t, 0, 2), color = red, key = "control var y(t)", explicit (sy, t, 0, 2), color = brown, key = " co-state var p(t)", explicit (sp, t, 0, 2))$

(%t93)



Check the sufficient conditions for a maximum in J for Example 3.

1. The constraint function g = 8*y is linear in y and is always both concave and convex, and we hence don't need to worry about the sign of p(t), the co-state variable.

2. We need the integrand f(x,y) of the functional J to be jointly concave in x and y. ConcaveTest (f, x, y) finds f(x,y) is jointly (simply) concave in both x and y, and the maximum is therefore a relative maximum.

(%i94) ConcaveTest (3*x - 2*y^2, x, y)$

$$d1 = \begin{pmatrix} 0 & 0 \\ 0 & -4 \end{pmatrix}$$

$d11 = 0$

$dd1 = 0$

$$d2 = \begin{pmatrix} -4 & 0 \\ 0 & 0 \end{pmatrix}$$

$d21 = -4$

$dd2 = 0$

simply concave

8.2 Prob. 21.4

Maximize the functional J = integrate (8*x - 10*y^2, t, 0, 4)]
subject to: dx/dt = 24*y, x(0) = 7, x(4) = free and unrestricted.

A. Set up the Hamiltonian, H = f + p*g.

(%i95) H : 8*x - 10*y^2 + p*24*y;

(H) $-10\,y^2+24\,p\,y+8\,x$

B. Assuming an interior solution, apply the maximum principle (1 & 2):
1. ∂H/∂y = 0:

(%i96) dy : diff (H, y);

(dy) $24\,p-20\,y$

(%i97) sy : solve(dy,y);

(sy) $[y=\frac{6\,p}{5}]$

sy is y(t) solution in terms of p(t).

(%i98) sy : rhs (sy[1]);

(sy) $\frac{6\,p}{5}$

2. use dx/dt = diff(H,p) and dp/dt = -diff(H,x).

(%i99) xd : diff(H,p);

(xd) $24\,y$

(%i100) xd : at (xd, y = sy);

(xd) $\frac{144\,p}{5}$

dx/dt depends on p(t).

(%i101) pd : - diff (H, x);

(pd) -8

We can immediately integrate this first order ode:
dp/dt = -8,
to get p(t) in terms of an arbitrary constant k1.

sp is p(t) in terms of constant k1.

```
(%i102) sp : integrate(pd, t) + k1;
```

(sp) $k1-8\,t$

replace p(t) in expression dx/dt (xd).

```
(%i103) xd : at(xd, p = sp), expand;
```

(xd) $\frac{144\,k1}{5}-\frac{1152\,t}{5}$

We can then integrate dx/dt = F(t) in terms of another arbitrary integration constant k2.
sx is x(t) in terms of k1 and k2.

```
(%i104) sx : integrate (xd, t) + k2;
```

(sx) $-\frac{576\,t^2}{5}+\frac{144\,k1\,t}{5}+k2$

Apply the boundary condtions at t = 0 and t = 4 to determine the values of k1 and k2.
Instead of x(4) value we use p(4) = 0 (transversality condition).

```
(%i105) ksoln : solve ([ at(sx, t = 0) = 7, at (sp, t = 4) = 0] );
```

(ksoln) $[[k1=32,k2=7]]$

```
(%i106) ksoln : ksoln[1];
```

(ksoln) $[k1=32,k2=7]$

Update sx, which is x(t).

```
(%i107) sx : at (sx, ksoln );
```

(sx) $-\frac{576\,t^2}{5}+\frac{4608\,t}{5}+7$

```
(%i108) float(sx);
```

(%o108) $-115.2\,t^2+921.6\,t+7.0$

```
(%i109) [ at (sx, t = 0), at (sx, t = 4) ];
```

(%o109) $[7,\frac{9251}{5}]$

```
(%i110) float(%);
```

(%o110) $[7.0,1850.2]$

Update sp which is p(t), the co-state variable.

```
(%i111) sp : at(sp, ksoln );
```

(sp) $32-8\,t$

```
(%i112) [at (sp, t = 0), at (sp, t = 4)];
```

(%o112) [32,0]

sy is y(t), the control variable, which is proportional to p, so update sy:

```
(%i113) sy : at (sy, p = sp), expand;
```

(sy) $\frac{192}{5}-\frac{48\,t}{5}$

```
(%i114) float(sy);
```

(%o114) $38.4-9.6\,t$

```
(%i115) at (sy, t = 4);
```

(%o115) 0

The optimal path of the control variable y(t) is linear, starting at (0, 38.4) and ending at (4, 0), with a slope of - 9.6.

```
(%i116) wxdraw2d (xlabel = "t", key = " state var x(t)", yrange = [0, 2e3], key_pos = top_left,
    explicit (sx, t, 0, 4), color = red, key = "control var y(t)", explicit (sy, t, 0, 4),
    color = brown, key = " co-state var p(t)", explicit (sp, t, 0, 4))$
```

(%t116)



Check the sufficient conditions for a maximum in J for Prob. 21.4.

1. The constraint function g = 24*y is linear in y and is always both concave and convex, and we hence don't need to worry about the sign of p(t), the co-state variable.

2. We need the integrand f(x,y) of the functional J to be jointly concave in x and y. ConcaveTest (f, x, y) finds f(x,y) is jointly (simply) concave in both x and y, and the maximum is therefore a relative maximum.

```
(%i117) ConcaveTest (8*x - 10*y^2, x, y)$
```

$$d1=\begin{pmatrix}0 & 0\\ 0 & -20\end{pmatrix}$$

$d11=0$

$dd1=0$

$$d2=\begin{pmatrix}-20 & 0\\ 0 & 0\end{pmatrix}$$

$d21=-20$

$dd2=0$

simply concave

8.3 Prob. 21.6 Use of Zindef_float (A, B)

Maximize the functional J = integrate (4*y - y^2 - x - 2*x^2, t, 0, 1)
subject to: dx/dt = x + y, x(0) = 6.15, x(1) = free and unrestricted.

A. Set up the Hamiltonian, H = f + p*g.

```
(%i118) H : 4*y - y^2 - x - 2*x^2 + p*(x + y);
```

(H) $-y^2+p\,(y+x)+4\,y-2\,x^2-x$

B. Assuming an interior solution, apply the maximum principle (1 & 2):
1. ∂H/∂y = 0:

```
(%i119) dy : diff (H, y);
```

(dy) $-2\,y+p+4$

```
(%i120) sy : solve(dy,y), expand;
```

(sy) $[y=\frac{p}{2}+2]$

sy is y(t) solution in terms of p(t).

(%i121) sy : rhs (sy[1]);

(sy) $\frac{p}{2}+2$

2. use dx/dt = diff(H,p) and dp/dt = -diff(H,x).

(%i122) xd : diff(H,p);

(xd) $y+x$

(%i123) xd : at (xd, y = sy);

(xd) $x+\frac{p}{2}+2$

dx/dt depends on both x(t) and p(t).

(%i124) pd : - diff (H, x);

(pd) $4\,x-p+1$

dp/dt depends on both x(t) and p(t).
So we have a pair of coupled first order odes to solve.
Let Z = cvec ([x, p]) and write the set of first order odes in matrix form:

dZ/dt = A . Z + B.

The Maxima function Zindef_float (A, B), defined in Econ2.mac, returns the indefinite solution as a matrix column vector.

(%i126) A : matrix ([1, 1/2], [4, -1]);
B : cvec ([2, 1]);

(A) $\begin{pmatrix} 1 & \frac{1}{2} \\ 4 & -1 \end{pmatrix}$

(B) $\begin{pmatrix} 2 \\ 1 \end{pmatrix}$

(%i127) eigenvalues(A);

(%o127) $[[-\sqrt{3},\sqrt{3}],[1,1]]$

(%i128) float(%);

(%o128) $[[-1.7321,1.7321],[1.0,1.0]]$

(%i129) Zs : Zindef_float (A, B);

(Zs) $$\begin{pmatrix} 1.0\ \%k_2\ \%e^{1.7321\,t}+1.0\ \%k_1\ \%e^{-1.7321\,t}-\frac{5}{6} \\ 1.4641\ \%k_2\ \%e^{1.7321\,t}-5.4641\ \%k_1\ \%e^{-1.7321\,t}-\frac{7}{3} \end{pmatrix}$$

(%i131) xs : Zs[1,1];
ps : Zs[2,1];

(xs) $$1.0\ \%k_2\ \%e^{1.7321\,t}+1.0\ \%k_1\ \%e^{-1.7321\,t}-\frac{5}{6}$$

(ps) $$1.4641\ \%k_2\ \%e^{1.7321\,t}-5.4641\ \%k_1\ \%e^{-1.7321\,t}-\frac{7}{3}$$

To determine the constants %k1 and %k2 we use x(0) = 6.15 and p(1) = 0, the latter corresponding to x(1) = free and unrestricted.

(%i132) ksoln : solve ([at(xs, t = 0) = 6.15, at (ps, t = 1) = 0]), numer;

(ksoln) $[[\%k_2=0.98291, \%k_1=6.0004]]$

(%i134) sx : at (xs, ksoln[1]);
sp : at (ps, ksoln[1]);

(sx) $$0.98291\ \%e^{1.7321\,t}+6.0004\ \%e^{-1.7321\,t}-\frac{5}{6}$$

(sp) $$1.4391\ \%e^{1.7321\,t}-32.787\ \%e^{-1.7321\,t}-\frac{7}{3}$$

To compare with Dowling's solutions we need to replace some pieces of this with floating point values, which we can do with the Maxima function substpart (zz, expr, n1, n2,...).

(%i135) part (sp, 3);

(%o135) $$-\frac{7}{3}$$

(%i136) sp : substpart (float (-7/3), sp, 3);

(sp) $$1.4391\ \%e^{1.7321\,t}-32.787\ \%e^{-1.7321\,t}-2.3333$$

(%i137) [at (sp, t = 0), at (sp, t = 1)];

(%o137) $[-33.681, -1.7764\ 10^{-14}]$

(%i138) sx : substpart (float (- 5/6), sx, 3);

(sx) $$0.98291\ \%e^{1.7321\,t}+6.0004\ \%e^{-1.7321\,t}-0.83333$$

(%i139) at (sx, t = 1);

(%o139) 5.7839

Finally, we can update sy, which is y(t), and which depends on p(t):

(%i140) sy : at (sy, p = sp), expand;

(sy) $0.71954\ \%e^{1.7321\ t} - 16.393\ \%e^{-1.7321\ t} + 0.83333$

(%i141) [at (sy, t = 0), at (sy, t = 1)];

(%o141) [−14.841,2.0]

(%i142) wxdraw2d (xlabel = "t", key = " state var x(t)", yrange = [-40, 20], key_pos = top_left, explicit (sx, t, 0, 1), color = red, key = "control var y(t)", explicit (sy, t, 0, 1), color = brown, key = " co-state var p(t)", explicit (sp, t, 0, 1), color = black, key = "", line_width = 1, explicit (0, t, 0, 1))$

(%t142)



Check the sufficient conditions for a maximum in J for Prob. 21.6.

1. The constraint function g = x + y is linear in x and y and is always both concave and convex, and hence we don't need to worry about the sign of p(t), the co-state variable.

2. We need the integrand f(x,y) of the functional J to be jointly concave in x and y. ConcaveTest (f, x, y) finds f(x,y) is strictly concave, and the maximum is therefore a global maximum.

(%i143) ConcaveTest (4*y - y^2 - x - 2*x^2, x, y)$

$$d1 = \begin{pmatrix} -4 & 0 \\ 0 & -2 \end{pmatrix}$$

$d11 = -4$

$dd1 = 8$

strictly concave

9 Case of Endpoint Inequality Constraints

Quoting Dowling:
"If the terminal value of the state variable is subject to an inequality constraint, x(T) >= xmin, the optimal value x_opt(T) may be chosen freely as long as it does not violate the value set by the constraint xmin. If x_opt(T) > xmin, the constraint is nonbinding and the problem reduces to a free endpoint problem. So
p(T) = 0 when x_opt(T) > xmin.

If x_opt(T) < xmin, the constraint is binding and the optimal solution will involve setting x(T) = xmin, which is equivalent to a fixed-end problem with
p(T) >= 0 when x_opt(T) = xmin.

For conciseness, the endpoint conditions are sometimes reduced to a single statement analogous to the Kuhn-Tucker condition:
p(T) >= 0, x(T) >= xmin, and [x(T) - xmin]*p(T) = 0."

"In practice, solving problems with inequality constraints on the endpoints is straightforward.

First solve the problem as if it were a free endpoint problem. If the optimal value of the state variable is greater than the minimum required by the endpoint condition, i.e., if x_opt(T) >= xmin, the correct solution has been found. If x_opt(T) < xmin, set the terminal endpoint equal to the value of the constraint, x(T) = xmin, and solve as a fixed endpoint problem."

9.1 Example 5

Maximize J = integrate (3*x - 2*y^2, t, 0, 2)
subject to: dx/dt = 8*y, x(0) = 5, x(2) >= 95.

To solve an optimal control problem involving an inequality constraint, solve it first as an unconstrained problem with a free end point. This we did in Example 3 where we found x_opt(t) = - 24*t^2 + 96*t + 5,

(%i144) at (-24*t^2 + 96*t + 5, t = 2);

(%o144) 101

Since x_opt(2) = 101 > 95, the free endpoint solution satisfies the inequality constraint required, and the constraint is nonbinding, so we have found the correct solution for the state variable x. The solution for the control variable y(t) found in Ex. 3 also is still valid:
y(t) = 12 - 6*t.

9.2 Example 6

Redo the same problem in Ex. 5 above with the new required boundary conditions:
x(0) = 5, x(2) >= 133.

Step 1 is to solve Ex. 5 as an unconstrained problem with a free endpoint, which we did in Ex. 3, finding x_opt(2) = 101, which is less than the required value of 133. Hence we need to solve as a fixed end point problem with x(2) = 133.

From our work above on Ex. 3, we found y = 2*p, dx/dt = 16*p, dp/dt = -3, p = k1 - 3*t, dx/dt = 16*k1 - 48*t, sx = -24*t^2 + 16*k1*t + k2.

```
(%i146) sx : -24*t^2 + 16*k1*t + k2;
        ksoln : solve ([ at(sx, t = 0) = 5, at (sx, t = 2) = 133])[1];
```

(sx) $-24\,t^2+16\,k1\,t+k2$

(ksoln) $[k1=7, k2=5]$

```
(%i147) sx : at (sx, ksoln);
```

(sx) $-24\,t^2+112\,t+5$

```
(%i148) [at (sx, t = 0),  at(sx, t = 2)];
```

(%o148) $[5, 133]$

```
(%i149) sp : at(k1 - 3*t, ksoln);
```

(sp) $7-3\,t$

```
(%i150) [at (sp, t = 0),  at (sp, t = 2)];
```

(%o150) $[7, 1]$

```
(%i151) sy : expand (2*sp);
```

(sy) $14-6\,t$

```
(%i152) [ at (sy, t = 0),  at (sy, t = 2)];
```

(%o152) $[14, 2]$

9.3 Example 7

Quoting Dowling:
"With an inequality constraint as a terminal endpoint...we first optimize the Hamiltonian subject to a free endpoint. With a free endpoint we set p(T) = 0, allowing the marginal value of the state variable to be taken down to zero. This, in effect, means that as long as the minimum value set by the constraint is met, the state variable is no longer of any value to us. Our interest in the state variable does not extend beyond time T."

"Most variables have value, however, and our interest generally extends beyond some narrowly limited time horizon. In such cases we will not treat the state variable as a free good by permitting its marginal value to be reduced to zero. We will rather require some minimum value of the state variable to be preserved for use beyond time T. This means maximizing the Hamiltonian subject to a fixed endpoint determined by the minimum value of the constraint. In such cases, p(T) > 0, the constraint is binding, and we will not use as much of the state variable as we would if it were a free good."

9.4 Prob. 21.7

Maximize J = integrate (8*x - 10*y^2, t, 0, 4)
subject to: dx/dt = 24*y, x(0) = 7, x(4) >= 2000.

This problem was previously solve in Prob. 21.4 as a free endpoint, where we found
y = 6*p/5, dx/dt = 144*p/5, dp/dt = -8, p = k1 - 8*t, dx/dt = 144*k1/5 - 1152*t/5,
x = - 576*t^2/5 + 144*k1*t/5 + k2, and with p(4) = 0, we got sx = -115.2*t^2 + 921.6*t + 7,
at(sx, t = 4) = 1850.2 < 2000. So the free endpoint solution does not provide a large enough value for x(4). We instead use x(0) = 7 and x(4) = 2000 to determine the constants k1 and k2.

```
(%i154) sx : - 576*t^2/5 + 144*k1*t/5 + k2;
        ksoln : solve ([ at (sx, t = 0) = 7,  at (sx, t = 4) = 2000] )[1], numer;
```

(sx) $-\frac{576\,t^2}{5}+\frac{144\,k1\,t}{5}+k2$

(ksoln) $[k1=33.3, k2=7]$

```
(%i155) sx : at (sx, ksoln), numer;
```

(sx) $-115.2\,t^2+959.05\,t+7$

```
(%i156) sp : at (k1 - 8*t, ksoln);
```

(sp) $33.3-8\,t$

```
(%i157) at (sp, t = 4);
```

(%o157) 1.3003

```
(%i158) sy : float (6*sp/5), expand;
```

(sy) $39.96-9.6\,t$

(%i159) at (sy, t = 4);

(%o159) 1.5604

9.5 Prob. 21.10 Use of Zindef_float (A, B)

Maximize J = integrate (4*y - y^2 - x - 2*x^2, t, 0, 1)
subject to dx/dt = x + y, x(0) = 6.15, x(1) >= 8.

We have already optimized this function subject to free endpoint conditions in Prob. 21.6 above. There we found for the value of x(1):

(%i160) float(0.98291*exp(sqrt(3)) + 6.0004*exp(-sqrt(3)) - 0.83333);

(%o160) 5.7839

which is less than 8, so we must evaluate the constants k1 and k2 using fixed endpoint conditions x(0) = 6.15, x(1) = 8.

Since the path to a solution involved the pair of coupled first order ode's dx/dt = u(x,p) and dp/dt = v(x,p) and we used matrix methods, we repeat our solution from scratch.

(%i162) H : 4*y - y^2 - x - 2*x^2 + p*(x + y);
dy : diff (H, y);

(H) $-y^2+p\,(y+x)+4\,y-2\,x^2-x$

(dy) $-2\,y+p+4$

(%i163) solve (dy, y);

(%o163) $[y=\frac{p+4}{2}]$

(%i164) sy : rhs(%[1]);

(sy) $\frac{p+4}{2}$

(%i165) xd : diff (H, p);

(xd) $y+x$

(%i166) xd : at (xd, y = sy), expand;

(xd) $x+\frac{p}{2}+2$

(%i167) pd : - diff (H, x);

(pd) $4\,x-p+1$

dp/dt depends on both x(t) and p(t), as does dx/dt.
So we have a pair of coupled first order odes to solve.
Let Z = cvec ([x, p]) and write the set of first order odes in matrix form:
dZ/dt = A . Z + B.
The Maxima function Zindef (A, B) returns the indefinite solution as a matrix column vector.

```
(%i169) A : matrix ( [1, 1/2], [4, -1] );
        B : cvec ( [2, 1]);
```

(A) $\begin{pmatrix} 1 & \frac{1}{2} \\ 4 & -1 \end{pmatrix}$

(B) $\begin{pmatrix} 2 \\ 1 \end{pmatrix}$

```
(%i170) eigenvalues (A);
```

(%o170) $[[-\sqrt{3},\sqrt{3}],[1,1]]$

```
(%i171) float (%);
```

(%o171) $[[-1.7321,1.7321],[1.0,1.0]]$

```
(%i172) Zs : Zindef_float (A, B);
```

(Zs) $\begin{pmatrix} 1.0\ \%k_2\ \%e^{1.7321\,t}+1.0\ \%k_1\ \%e^{-1.7321\,t}-\frac{5}{6} \\ 1.4641\ \%k_2\ \%e^{1.7321\,t}-5.4641\ \%k_1\ \%e^{-1.7321\,t}-\frac{7}{3} \end{pmatrix}$

```
(%i174) xs : Zs[1,1];
        ps : Zs[2,1];
```

(xs) $1.0\ \%k_2\ \%e^{1.7321\,t}+1.0\ \%k_1\ \%e^{-1.7321\,t}-\frac{5}{6}$

(ps) $1.4641\ \%k_2\ \%e^{1.7321\,t}-5.4641\ \%k_1\ \%e^{-1.7321\,t}-\frac{7}{3}$

To determine the constants %k1 and %k2 we use x(0) = 6.15 and x(1) = 8.

```
(%i175) ksoln : solve ([at(xs, t = 0) = 6.15, at (xs, t = 1) = 8]);
```

(ksoln) $[[\%k_2=\frac{3802758981311531 09}{274042507698401220},\%k_1=\frac{3833635701573371 86}{685106269246003 05}]]$

```
(%i176) ksoln : float (ksoln);
```

(ksoln) $[[\%k_2=1.3877,\%k_1=5.5957]]$

(%i178) sx : at (xs, ksoln[1]);
sp : at (ps, ksoln[1]);

(sx) $1.3877\ \%e^{1.7321\,t}+5.5957\ \%e^{-1.7321\,t}-\frac{5}{6}$

(sp) $2.0317\ \%e^{1.7321\,t}-30.575\ \%e^{-1.7321\,t}-\frac{7}{3}$

To compare with Dowling's solutions we need to replace some pieces of this with floating point values, which we can do with the Maxima function substpart (zz, expr, n1, n2,...).

(%i179) part (sp, 3);

(%o179) $-\frac{7}{3}$

(%i180) sp : substpart (float (-7/3), sp, 3);

(sp) $2.0317\ \%e^{1.7321\,t}-30.575\ \%e^{-1.7321\,t}-2.3333$

(%i181) [at (sp, t = 0), at (sp, t = 1)];

(%o181) [−30.877,3.7407]

(%i182) sx : substpart (float (- 5/6), sx, 3);

(sx) $1.3877\ \%e^{1.7321\,t}+5.5957\ \%e^{-1.7321\,t}-0.83333$

Check x(1) value:

(%i183) at (sx, t = 1);

(%o183) 8.0

Finally, we can update sy, which is y(t), and which depends on p(t):

(%i184) sy : at (sy, p = sp), expand;

(sy) $1.0158\ \%e^{1.7321\,t}-15.288\ \%e^{-1.7321\,t}+0.83333$

(%i185) [at (sy, t = 0), at (sy, t = 1)];

(%o185) [−13.439,3.8703]

(%i186) wxdraw2d (xlabel = "t", key = " state var x(t)", yrange = [-40, 20], key_pos = top_left,
explicit (sx, t, 0, 1), color = red, key = "control var y(t)", explicit (sy, t, 0, 1),
color = brown, key = " co-state var p(t)", explicit (sp, t, 0, 1), color = black,
key = "", line_width = 1, explicit (0, t, 0, 1))$

(%t186)



Check the sufficient conditions for a maximum in J for Prob. 21.10.
1. The constraint function g = x + y is linear in x and y and is always both concave and convex, and hence we don't need to worry about the sign of p(t), the co-state variable.
2. We need the integrand f(x,y) of the functional J to be jointly concave in x and y.
ConcaveTest (f, x, y) finds f(x,y) is strictly concave, and the maximum is therefore a global maximum.

(%i187) ConcaveTest (4*y - y^2 - x - 2*x^2, x, y)$

$$d1=\begin{pmatrix}-4 & 0\\ 0 & -2\end{pmatrix}$$

$d11=-4$

$dd1=8$

strictly concave

10 The Current-Valued Hamiltonian Hc

10.1 Modified Maximum Principle with Hc

We seek to maximize J = integrate (exp(-r*t)*f(x(t),y(t),t), t, 0, T), with r = constant,
subject to: dx/dt = g(x(t), y(t), t), x(0) = x0, x(T) = free.
Using p(t) as the co-state variable again, the Hamiltonian is
H = exp (-r*t)*f(x,y,t) + p(t)*g(x,y,t).

Let μ = p*exp (r*t), then p = μ*exp(-r*t) and ∂μ/∂p = exp(r*t).

Now define the current valued Hamiltonian Hc as:
Hc = H*exp(r*t) = f + μ*g, and then, of course, H = Hc*exp(-r*t).

Then dp/dt = -∂H/∂x = - ∂/∂x (Hc*exp(-r*t)) = - exp(-r*t)*∂Hc/∂x.

But using p = μ*exp(-r*t), we also have
dp/dt = -r*μ*exp(-r*t) + exp(-r*t)*dμ/dt = - exp(-r*t)*∂Hc/∂x.

Cancelling the common factor exp (-r*t), we get:

dμ/dt = r*μ - ∂Hc/∂x.

Moreover, the endpoint condition p(T) = 0 implies μ(T)*exp(-r*T) = 0.

Assuming an interior solution, the requirement
∂H/∂y = 0 implies ∂(Hc*exp(-r*t))/∂y = 0, or, because exp(-r*t) > 0,

∂Hc/∂y = 0

We next work on dx/dt.
dx/dt = ∂H/∂p = ∂ (Hc*exp(-r*t)) /∂p = exp(-r*t)*∂Hc/∂p = exp(-r*t)*∂μ/∂p*∂Hc/∂μ,
using the chain rule of differentiation. But ∂μ/∂p = exp(r*t), so we finally have

dx/dt = ∂Hc/∂μ.

Summarizing the current-valued Hamiltonian approach to the first order equations of motion:

1. For an interior solution, require ∂Hc/∂y = 0.
2. Require the pair of equations:
   dx/dt = ∂Hc/∂μ, dμ/dt = r*μ - ∂Hc/∂x.
3. For x(T) = free boundary condition, use μ(T)*exp(-r*T) = 0.

10.2 Example 8 Use of Zindef_float (A, B)

Maximize J = integrate (exp (-0.02*t)*(x - 3*x^2 - 2*y^2), t, 0, 2)
subject to: dx/dt = y - 0.5*x, x(0) = 93.91, x(2) = free.

A. Set up the current valued Hamiltonian Hc

(%i188) Hc : x - 3*x^2 - 2*y^2 + μ*(y - 0.5*x);

(Hc) $(y-0.5\,x)\,\mu-2\,y^2-3\,x^2+x$

B. Assuming an interior solution, apply the modified Maximum principle.
1. ∂Hc/∂y = 0.

(%i189) dy : diff (Hc, y);

(dy) $\mu-4\,y$

(%i190) solve (dy, y);

(%o190) $[y=\frac{\mu}{4}]$

(%i191) sy : rhs (%[1]);

(sy) $\frac{\mu}{4}$

2. dx/dt = ∂Hc/∂μ and dμ/dt = r*μ - ∂Hc/∂x:

(%i192) xd : diff (Hc, μ);

(xd) $y-0.5\,x$

(%i193) xd : at (xd, y = sy), numer;

(xd) $0.25\,\mu-0.5\,x$

(%i194) μd : 0.02*μ - diff (Hc, x);

(μd) $0.52\,\mu+6\,x-1$

dx/dt and dμ/dt each depend on x and μ, so we have a pair of first order ode's which we solve using matrix methods, as in Prob. 21.6 and Prob. 21.10 above. Our solution matrix column vector Zs has the form:

(%i195) cvec([μ, x]);

(%o195) $\begin{pmatrix}\mu\\x\end{pmatrix}$

(%i197) A : matrix ([0.52, 6], [0.25, -0.5]);
B : cvec ([-1, 0]);

(A) $\begin{pmatrix} 0.52 & 6 \\ 0.25 & -0.5 \end{pmatrix}$

(B) $\begin{pmatrix} -1 \\ 0 \end{pmatrix}$

(%i198) eigenvalues (A);

(%o198) $[[-\frac{\sqrt{17601}-1}{100},\frac{\sqrt{17601}+1}{100}],[1,1]]$

(%i199) float(%);

(%o199) $[[-1.3167,1.3367],[1.0,1.0]]$

(%i200) cvec ([dμ/dt, dx/dt]) = A . cvec ([μ, x]) + B;

(%o200) $\begin{pmatrix} \frac{d\mu}{dt} \\ \frac{dx}{dt} \end{pmatrix}=\begin{pmatrix} 0.52\ \mu+6\ x-1 \\ 0.25\ \mu-0.5\ x \end{pmatrix}$

(%i201) Zs : Zindef_float (A, B);

(Zs) $\begin{pmatrix} 1.0\ \%k_2\ \%e^{1.3367\ t}+1.0\ \%k_1\ \%e^{-1.3167\ t}+0.28409 \\ 0.13611\ \%k_2\ \%e^{1.3367\ t}-0.30611\ \%k_1\ \%e^{-1.3167\ t}+0.14205 \end{pmatrix}$

(%i203) μs : Zs[1,1];
xs : Zs[2,1];

(μs) $1.0\ \%k_2\ \%e^{1.3367\ t}+1.0\ \%k_1\ \%e^{-1.3167\ t}+0.28409$

(xs) $0.13611\ \%k_2\ \%e^{1.3367\ t}-0.30611\ \%k_1\ \%e^{-1.3167\ t}+0.14205$

With T = 2, we require exp(-r*T)*μ(T) = 0 as part of the next step.

(%i204) ksoln : float (solve ([at (xs, t = 0) = 93.91, exp(-0.02*2)*at (μs, t = 2) = 0]));

(ksoln) $[[\%k_2=1.4958,\%k_1=-305.65]]$

(%i205) ksoln : ksoln[1];

(ksoln) $[\%k_2=1.4958,\%k_1=-305.65]$

(%i207) sμ : at(μs, ksoln);
sx : at (xs, ksoln);

(sμ) $1.4958\ \%e^{1.3367\, t}-305.65\ \%e^{-1.3167\, t}+0.28409$

(sx) $0.2036\ \%e^{1.3367\, t}+93.564\ \%e^{-1.3167\, t}+0.14205$

(%i208) [at (sx, t = 0), at (sx, t = 2)];

(%o208) $[93.91, 9.8132]$

(%i209) [at (sμ, t = 0), at (sμ, t = 2)];

(%o209) $[-303.87, 2.8422\ 10^{-14}]$

(%i210) sy : at (sy, μ = sμ), expand;

(sy) $0.37395\ \%e^{1.3367\, t}-76.413\ \%e^{-1.3167\, t}+0.071023$

(%i211) [at(sy, t = 0), at (sy, t = 2)];

(%o211) $[-75.968, 7.1054\ 10^{-15}]$

(%i212) wxdraw2d (xlabel = "t", key = " state var x(t)", yrange = [-350, 150], key_pos = bottom_right, explicit (sx, t, 0, 2), color = red, key = "control var y(t)", explicit (sy, t, 0, 2), color = brown, key = " co-state var mu(t)", explicit (sμ, t, 0, 2), color = black, key = "", line_width = 1, explicit (0, t, 0, 2))$

(%t212)



Applying the sufficiency test we look at the matrix

(%i213) matrix([Fxx, Fxy], [Fyx, Fyy]);

(%o213) $\begin{pmatrix} Fxx & Fxy \\ Fyx & Fyy \end{pmatrix}$

with F = exp(-r*t)*f(x,y) and with r a constant, so Fxy = exp(-r*t)*fxy, etc, the sign of Fxx will be the same as the sign of fxx, and the determinant of the Hessian matrix of F will be equal to exp(-2*r*t) times the determinant of the Hessian matrix of f, with both determinants having the same sign.

Hence we can use ConcaveTest(f(x,y), x, y) and ignore the exponential factor.

(%i214) ConcaveTest (x - 3*x^2 - 2*y^2, x, y)$

$$d1=\begin{pmatrix}-6 & 0\\ 0 & -4\end{pmatrix}$$

$d11=-6$

$dd1=24$

strictly concave

With g = x + y linear in both x and y, we therefore have found a global maximum.

10.3 Prob. 21.11 Use of Zindef_float (A,B)

Maximize J = integrate (exp(- 0.05*t)*(x*y - x^2 - y^2), t, 0, 3)
subject to: dx/dt = x + y, x(0) = 134.35, x(3) = free.

A. Set up the current valued Hamiltonian Hc

(%i215) Hc : x*y - x^2 - y^2 + μ*(x + y);

(Hc) $(y+x)\,\mu-y^2+x\,y-x^2$

B. Assuming an interior solution, apply the modified Maximum principle.
1. ∂Hc/∂y = 0.

(%i216) dy : diff (Hc, y);

(dy) $\mu-2\,y+x$

(%i217) solve (dy, y);

(%o217) $[y=\frac{\mu+x}{2}]$

(%i218) sy : rhs (%[1]);

(sy) $\frac{\mu+x}{2}$

(%i219) sy : float (expand (sy));

(sy) $0.5\,\mu+0.5\,x$

2. dx/dt = ∂Hc/∂μ and dμ/dt = r*μ - ∂Hc/∂x:

(%i220) xd : diff (Hc, μ);

(xd) $y+x$

(%i221) xd : at (xd, y = sy);

(xd) $0.5\,\mu+1.5\,x$

(%i222) μd : 0.05*μ - diff (Hc, x);

(μd) $-0.95\,\mu-y+2\,x$

(%i223) μd : at (μd, y = sy);

(μd) $1.5\,x-1.45\,\mu$

(%i225) A : matrix ([- 1.45, 1.5], [0.5, 1.5]);
B : cvec ([0, 0]);

(A) $\begin{pmatrix}-1.45 & 1.5\\ 0.5 & 1.5\end{pmatrix}$

(B) $\begin{pmatrix}0\\ 0\end{pmatrix}$

(%i226) eigenvalues (A);

(%o226) $[[-\frac{\sqrt{4681}-1}{40},\frac{\sqrt{4681}+1}{40}],[1,1]]$

(%i227) float(%);

(%o227) $[[-1.6854,1.7354],[1.0,1.0]]$

(%i228) cvec ([dμ/dt, dx/dt]) = A . cvec ([μ, x]) + B;

(%o228) $\begin{pmatrix}\frac{d\mu}{dt}\\ \frac{dx}{dt}\end{pmatrix}=\begin{pmatrix}1.5\,x-1.45\,\mu\\ 0.5\,\mu+1.5\,x\end{pmatrix}$

(%i229) Zs : Zindef_float (A, B);

(Zs) $\begin{pmatrix} 1.0\ \%k_2\ \%e^{1.7354\ t}+1.0\ \%k_1\ \%e^{-1.6854\ t} \\ 2.1236\ \%k_2\ \%e^{1.7354\ t}-0.15696\ \%k_1\ \%e^{-1.6854\ t} \end{pmatrix}$

(%i231) μs : Zs[1,1];
xs : Zs[2,1];

(μs) $1.0\ \%k_2\ \%e^{1.7354\ t}+1.0\ \%k_1\ \%e^{-1.6854\ t}$

(xs) $2.1236\ \%k_2\ \%e^{1.7354\ t}-0.15696\ \%k_1\ \%e^{-1.6854\ t}$

(%i232) ksoln : float (solve ([at (xs, t = 0) = 134.35, exp(-0.05*3)*at (μs, t = 3) = 0]));

(ksoln) $[[\%k_2=0.029868, \%k_1=-855.53]]$

(%i233) ksoln : ksoln[1];

(ksoln) $[\%k_2=0.029868, \%k_1=-855.53]$

(%i235) sμ : at(μs, ksoln);
sx : at (xs, ksoln);

(sμ) $0.029868\ \%e^{1.7354\ t}-855.53\ \%e^{-1.6854\ t}$

(sx) $0.063429\ \%e^{1.7354\ t}+134.29\ \%e^{-1.6854\ t}$

(%i236) [at (sx, t = 0), at (sx, t = 3)];

(%o236) [134.35,12.426]

(%i237) [at (sμ, t = 0), at (sμ, t = 3)];

(%o237) $[-855.5, 1.7764\ 10^{-15}]$

(%i238) sy : at (sy, [μ = sμ, x = sx]), expand;

(sy) $0.046649\ \%e^{1.7354\ t}-360.62\ \%e^{-1.6854\ t}$

(%i239) [at(sy, t = 0), at (sy, t = 3)];

(%o239) [−360.57,6.2131]

(%i240) wxdraw2d (xlabel = "t", key = " state var x(t)", yrange = [-900, 150], key_pos = bottom_right, explicit (sx, t, 0, 3), color = red, key = "control var y(t)", explicit (sy, t, 0, 3), color = brown, key = " co-state var mu(t)", explicit (sμ, t, 0, 3), color = black, key = "", line_width = 1, explicit (0, t, 0, 3))$

(%t240)



(%i241) ConcaveTest (x*y - x^2 - y^2, x, y)$

$$d1 = \begin{pmatrix} -2 & 1 \\ 1 & -2 \end{pmatrix}$$

$d11 = -2$

$dd1 = 3$

strictly concave

With g = x + y linear in both x and y, we therefore have found a global maximum.